NAG Library Routine Document

D02AGF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

D02AGF solves a two-point boundary value problem for a system of ordinary differential equations, using initial value techniques and Newton iteration; it generalizes D02HAF to include the case where parameters other than boundary values are to be determined.

2 Specification

```
 SUBROUTINE D02AGF(H, E, PARERR, PARAM, C, N, N1, M1, AUX, BCAUX, RAAUX,
1                  PRSOL, MAT, COPY, WSPACE, WSPAC1, WSPAC2, IFAIL)
 INTEGER           N, N1, M1, IFAIL
 double precision  H, E(N), PARERR(N1), PARAM(N1), C(M1,N), MAT(N1,N1),
1                  COPY(1,1), WSPACE(N,9), WSPAC1(N), WSPAC2(N)
 EXTERNAL          AUX, BCAUX, RAAUX, PRSOL
```

3 Description

D02AGF solves a two-point boundary value problem by determining the unknown parameters $p_1, p_2, \ldots, p_{n_1}$ of the problem. These parameters may be, but need not be, boundary values (as they are in D02HAF); they may include eigenvalue parameters in the coefficients of the differential equations, length of the range of integration, etc. The notation and methods used are similar to those of D02HAF and you are advised to study this first. (There the parameters $p_1, p_2, \ldots, p_{n_1}$ correspond to the unknown boundary conditions.) It is assumed that we have a system of n first-order ordinary differential equations of the form

$$\frac{dy_i}{dx} = f_i(x, y_1, y_2, \ldots, y_n), \qquad i = 1, 2, \ldots, n,$$

and that derivatives f_i are evaluated by AUX. The system, including the boundary conditions given by BCAUX, and the range of integration and matching point, r, given by RAAUX, involves the n_1 unknown parameters $p_1, p_2, \ldots, p_{n_1}$ which are to be determined, and for which initial estimates must be supplied. The number of unknown parameters n_1 must not exceed the number of equations n. If $n_1 < n$, we assume that $(n - n_1)$ equations of the system are not involved in the matching process. These are usually referred to as 'driving equations'; they are independent of the parameters and of the solutions of the other n_1 equations. In numbering the equations for AUX, the driving equations must be put last.

The estimated values of the parameters are corrected by a form of Newton iteration. The Newton correction on each iteration is calculated using a matrix whose (i, j)th element depends on the derivative of the ith component of the solution, y_i, with respect to the jth parameter, p_j. This matrix is calculated by a simple numerical differentiation technique which requires n_1 evaluations of the differential system.

4 References

None.

5 Parameters

You are strongly recommended to read Sections 3 and 8 in conjunction with this section.

1: H – ***double precision*** *Input/Output*

On entry: H must be set to an estimate of the step size, h, needed for integration.

On exit: the last step length used.

2: E(N) – ***double precision*** array *Input*

On entry: E(i) must be set to a small quantity to control the ith solution component. The element E(i) is used:

(i) in the bound on the local error in the ith component of the solution y_i during integration,

(ii) in the convergence test on the ith component of the solution y_i at the matching point in the Newton iteration.

The elements E(i) should not be chosen too small. They should usually be several orders of magnitude larger than ***machine precision***.

3: PARERR(N1) – ***double precision*** array *Input*

On entry: PARERR(i) must be set to a small quantity to control the ith parameter component. The element PARERR(i) is used:

(i) in the convergence test on the ith parameter in the Newton iteration,

(ii) in perturbing the ith parameter when approximating the derivatives of the components of the solution with respect to the ith parameter, for use in the Newton iteration.

The elements PARERR(i) should not be chosen too small. They should usually be several orders of magnitude larger than ***machine precision***.

4: PARAM(N1) – ***double precision*** array *Input/Output*

On entry: PARAM(i) must be set to an estimate for the ith parameter, p_i, for $i = 1, 2, \ldots, \text{N1}$.

On exit: the corrected value for the ith parameter, unless an error has occurred, when it contains the last calculated value of the parameter (possibly perturbed by $\text{PARERR}(i) \times (1 + |\text{PARAM}(i)|)$ if the error occurred when calculating the approximate derivatives).

5: C(M1,N) – ***double precision*** array *Output*

On exit: the solution when $\text{M1} > 1$ (see M1).

If $\text{M1} = 1$, the elements of C are not used.

6: N – INTEGER *Input*

On entry: n, the total number of differential equations.

7: N1 – INTEGER *Input*

On entry: n_1, the number of parameters.

If $\text{N1} < \text{N}$, the last $\text{N} - \text{N1}$ differential equations (in AUX) are driving equations (see Section 3).

Constraint: $\text{N1} \leq \text{N}$.

8: M1 – INTEGER *Input*

On entry: determines whether or not the final solution is computed as well as the parameter values.

$\text{M1} = 1$
The final solution is not calculated;

$\text{M1} > 1$
The final values of the solution at interval (length of range)/$(\text{M1} - 1)$ are calculated and stored sequentially in the array C starting with the values of y_i evaluated at the first end point (see RAAUX) stored in C$(1, i)$.

9: AUX – SUBROUTINE, supplied by the user. *External Procedure*

AUX must evaluate the functions f_i (i.e., the derivatives y_i') for given values of its arguments, $x, y_1, \ldots, y_n, p_1, \ldots, p_{n_1}$.

The specification of AUX is:

```
SUBROUTINE AUX(F, Y, X, PARAM)
double precision  F(n), Y(n), X, PARAM(n1)
```

where n and $n1$ are the numerical values of N and N1 in the call of D02AGF.

1: F(n) – ***double precision*** array *Output*

On exit: the value of f_i, for $i = 1, 2, \ldots, n$.

2: Y(n) – ***double precision*** array *Input*

On entry: y_i, the value of the argument, for $i = 1, 2, \ldots, n$.

3: X – ***double precision*** *Input*

On entry: x, the value of the argument.

4: PARAM($n1$) – ***double precision*** array *Input*

On entry: p_i, the value of the argument, for $i = 1, 2, \ldots, n_1$.

AUX must be declared as EXTERNAL in the (sub)program from which D02AGF is called. Parameters denoted as *Input* must **not** be changed by this procedure.

10: BCAUX – SUBROUTINE, supplied by the user. *External Procedure*

BCAUX must evaluate the values of y_i at the end points of the range given the values of $p_1, \ldots, p_{n_1}$.

The specification of BCAUX is:

```
SUBROUTINE BCAUX(G0, G1, PARAM)
double precision  G0(n), G1(n), PARAM(n1)
```

where n and $n1$ are the numerical values of N and N1 in the call of D02AGF.

1: G0(n) – ***double precision*** array *Output*

On exit: the values y_i, for $i = 1, 2, \ldots, n$, at the boundary point x_0 (see RAAUX).

2: G1(n) – ***double precision*** array *Output*

On exit: the values y_i, for $i = 1, 2, \ldots, n$, at the boundary point x_1 (see RAAUX).

3: PARAM($n1$) – ***double precision*** array *Input*

On entry: p_i, the value of the argument, for $i = 1, 2, \ldots, n$.

BCAUX must be declared as EXTERNAL in the (sub)program from which D02AGF is called. Parameters denoted as *Input* must **not** be changed by this procedure.

11: RAAUX – SUBROUTINE, supplied by the user. *External Procedure*

RAAUX must evaluate the end points, x_0 and x_1, of the range and the matching point, r, given the values $p_1, p_2, \ldots, p_{n_1}$.

The specification of RAAUX is:

```
SUBROUTINE RAAUX(X0, X1, R, PARAM)
double precision     X0, X1, R, PARAM(n1)
```

where $n1$ is the numerical value of N1 in the call of D02AGF.

1: X0 – ***double precision*** *Output*

On exit: must contain the left-hand end of the range, x_0.

2: X1 – ***double precision*** *Output*

On exit: must contain the right-hand end of the range x_1.

3: R – ***double precision*** *Output*

On exit: must contain the matching point, r.

4: PARAM($n1$) – ***double precision*** array *Input*

On entry: p_i, the value of the argument, for $i = 1, 2, \ldots, n_1$.

RAAUX must be declared as EXTERNAL in the (sub)program from which D02AGF is called. Parameters denoted as *Input* must **not** be changed by this procedure.

12: PRSOL – SUBROUTINE, supplied by the user. *External Procedure*

PRSOL is called at each iteration of the Newton method and can be used to print the current values of the parameters p_i, for $i = 1, 2, \ldots, n_1$, their errors, e_i, and the sum of squares of the errors at the matching point, r.

The specification of PRSOL is:

```
SUBROUTINE PRSOL(PARAM, RES, N1, ERR)
INTEGER              N1
double precision     PARAM(N1), RES, ERR(N1)
```

1: PARAM(N1) – ***double precision*** array *Input*

On entry: p_i, the current value of the parameters, for $i = 1, 2, \ldots, n_1$.

2: RES – ***double precision*** *Input*

On entry: the sum of squares of the errors in the parameters, $\sum_{i=1}^{n_1} e_i^2$.

3: N1 – INTEGER *Input*

On entry: n_1, the number of parameters.

4: ERR(N1) – ***double precision*** array *Input*

On entry: the errors in the parameters, e_i, for $i = 1, 2, \ldots, n_1$.

PRSOL must be declared as EXTERNAL in the (sub)program from which D02AGF is called. Parameters denoted as *Input* must **not** be changed by this procedure.

13: MAT(N1,N1) – ***double precision*** array *Workspace*
14: COPY(1,1) – ***double precision*** array *Input*
15: WSPACE(N,9) – ***double precision*** array *Workspace*
16: WSPAC1(N) – ***double precision*** array *Workspace*
17: WSPAC2(N) – ***double precision*** array *Workspace*

18: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, −1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value −1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value −1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL = 0 or −1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

This indicates that N1 > N on entry, that is the number of parameters is greater than the number of differential equations.

IFAIL = 2

As for IFAIL = 4 except that the integration failed while calculating the matrix for use in the Newton iteration.

IFAIL = 3

The current matching point r does not lie between the current end points x_0 and x_1. If the values x_0, x_1 and r depend on the parameters p_i, this may occur at any time in the Newton iteration if care is not taken to avoid it when coding RAAUX.

IFAIL = 4

The step length for integration H has halved more than 13 times (or too many steps were needed to reach the end of the range of integration) in attempting to control the local truncation error whilst integrating to obtain the solution corresponding to the current values p_i. If, on failure, H has the sign of $r - x_0$ then failure has occurred whilst integrating from x_0 to r, otherwise it has occurred whilst integrating from x_1 to r.

IFAIL = 5

The matrix of the equations to be solved for corrections to the variable parameters in the Newton method is singular (as determined by F03AFF).

IFAIL = 6

A satisfactory correction to the parameters was not obtained on the last Newton iteration employed. A Newton iteration is deemed to be unsatisfactory if the sum of the squares of the residuals (which can be printed using PRSOL) has not been reduced after three iterations using a new Newton correction.

IFAIL $= 7$

Convergence has not been obtained after 12 satisfactory iterations of the Newton method.

A further discussion of these errors and the steps which might be taken to correct them is given in Section 8.

7 Accuracy

If the process converges, the accuracy to which the unknown parameters are determined is usually close to that specified by you; and the solution, if requested, is usually determined to the accuracy specified.

8 Further Comments

The time taken by D02AGF depends on the complexity of the system, and on the number of iterations required. In practice, integration of the differential equations is by far the most costly process involved.

There may be particular difficulty in integrating the differential equations in one direction (indicated by IFAIL $= 2$ or 4). The value of r should be adjusted to avoid such difficulties.

If the matching point r is at one of the end points x_0 or x_1 and some of the parameters are used **only** to determine the boundary values at this point, then good initial estimates for these parameters are not required, since they are completely determined by the routine (for example, see p_2 in EX1 of Section 9).

Wherever they occur in the procedure, the error parameters contained in the arrays E and PARERR are used in 'mixed' form; that is E(i) always occurs in expressions of the form E$(i) \times (1 + |y_i|)$, and PARERR(i) always occurs in expressions of the form PARERR$(i) \times (1 + |p_i|)$. Though not ideal for every application, it is expected that this mixture of absolute and relative error testing will be adequate for most purposes.

Note that **convergence is not guaranteed**. You are strongly advised to provide an output PRSOL, as shown in EX1 of Section 9, in order to monitor the progress of the iteration. Failure of the Newton iteration to converge (see IFAIL $= 6$ or 7) usually results from poor starting approximations to the parameters, though occasionally such failures occur because the elements of one or both of the arrays PARERR or E are too small. (It should be possible to distinguish these cases by studying the output from PRSOL.) Poor starting approximations can also result in the failure described under IFAIL $= 4$ and 5 in Section 6 (especially if these errors occur after some Newton iterations have been completed, that is, after two or more calls of PRSOL). More frequently, a singular matrix in the Newton method (monitored as IFAIL $= 5$) occurs because the mathematical problem has been posed incorrectly. The case IFAIL $= 4$ usually occurs because h or r has been poorly estimated, so these values should be checked first. If IFAIL $= 2$ is monitored, the solution $y_1, y_2, \ldots, y_n$ is sensitive to perturbations in the parameters p_i. Reduce the size of one or more values PARERR(i) to reduce the perturbations. Since only one value p_i is perturbed at any time when forming the matrix, the perturbation which is too large can be located by studying the final output from PRSOL and the values of the parameters returned by D02AGF. If this change leads to other types of failure improve the initial values of p_i by other means.

The computing time for integrating the differential equations can sometimes depend critically on the quality of the initial estimates for the parameters p_i. If it seems that too much computing time is required and, in particular, if the values ERR(i) (available on each call of PRSOL) are much larger than the expected values of the solution at the matching point r, then the coding of AUX, BCAUX and RAAUX should be checked for errors. If no errors can be found, an independent attempt should be made to improve the initial estimates for PARAM(i).

The subroutine can be used to solve a very wide range of problems, for example:

(a) eigenvalue problems, including problems where the eigenvalue occurs in the boundary conditions;

(b) problems where the differential equations depend on some parameters which are to be determined so as to satisfy certain boundary conditions (see EX1 in Section 9);

(c) problems where one of the end points of the range of integration is to be determined as the point where a variable y_i takes a particular value (see EX2 in Section 9);

(d) singular problems and problems on infinite ranges of integration where the values of the solution at x_0 or x_1 or both are determined by a power series or an asymptotic expansion (or a more complicated expression) and where some of the coefficients in the expression are to be determined (see EX1 in Section 9); and

(e) differential equations with certain terms defined by other independent (driving) differential equations.

9 Example

For this routine two examples are presented. There is a single example program for D02AGF, with a main program and the code to solve the two example problems given in Example 1 (EX1) and Example 2 (EX2).

Example 1 (EX1)

This example finds the solution of the differential equation

$$y'' = \frac{y^3 - y'}{2x}$$

on the range $0 \leq x \leq 16$, with boundary conditions $y(0) = 0.1$ and $y(16) = 1/6$.

We cannot use the differential equation at $x = 0$ because it is singular, so we take the truncated series expansion

$$y(x) = \tfrac{1}{10} + p_1\frac{\sqrt{x}}{10} + \frac{x}{100}$$

near the origin (which is correct to the number of terms given in this case). Here p_1 is one of the parameters to be determined. We choose the range as $[0.1, 16]$ and setting $p_2 = y'(16)$, we can determine all the boundary conditions. We take the matching point to be 16, the end of the range, and so a good initial guess for p_2 is not necessary. We write $y = \mathrm{Y}(1)$, $y' = \mathrm{Y}(2)$, and estimate $p_1 = \mathrm{PARAM}(1) = 0.2$, $p_2 = \mathrm{PARAM}(2) = 0.0$.

Example 2 (EX2)

This example finds the gravitational constant p_1 and the range p_2 over which a projectile must be fired to hit the target with a given velocity. The differential equations are

$$y' = \tan\phi$$

$$v' = \frac{-\left(p_1 \sin\phi + 0.00002v^2\right)}{v\cos\phi}$$

$$\phi' = \tfrac{-p_1}{v^2}k$$

on the range $0 < x < p_2$ with boundary conditions

$$\begin{aligned} y &= 0, \quad v = 500, \quad \phi = 0.5 \quad \text{at} \quad x = 0 \\ y &= 0, \quad v = 450, \quad \phi = p_3 \quad \text{at} \quad x = p_2. \end{aligned}$$

We write $y = \mathrm{Y}(1)$, $v = \mathrm{Y}(2)$, $\phi = \mathrm{Y}(3)$, and we take the matching point $r = p_2$. We estimate $p_1 = \mathrm{PARAM}(1) = 32$, $p_2 = \mathrm{PARAM}(2) = 6000$ and $p_3 = \mathrm{PARAM}(3) = 0.54$ (though this estimate is not important).

9.1 Program Text

```
*     D02AGF Example Program Text
*     Mark 14 Revised. NAG Copyright 1989.
*     .. Parameters ..
      INTEGER          NOUT
      PARAMETER        (NOUT=6)
*     .. External Subroutines ..
      EXTERNAL         EX1, EX2
*     .. Executable Statements ..
      WRITE (NOUT,*) 'D02AGF Example Program Results'
      CALL EX1
```

```
      CALL EX2
      END
*
      SUBROUTINE EX1
*     .. Parameters ..
      INTEGER          N, M1
      PARAMETER        (N=2,M1=6)
      INTEGER          NOUT
      PARAMETER        (NOUT=6)
*     .. Scalars in Common ..
      INTEGER          IPRINT
*     .. Local Scalars ..
      DOUBLE PRECISION DUM, H, R, X, X1
      INTEGER          I, IFAIL, J, N1
*     .. Local Arrays ..
      DOUBLE PRECISION C(M1,N), DUMMY(1,1), E(N), MAT(N,N), PARAM(N),
     +                 PARERR(N), WSPAC1(N), WSPAC2(N), WSPACE(N,9)
*     .. External Subroutines ..
      EXTERNAL         AUX1, BCAUX1, D02AGF, PRSOL, RNAUX1
*     .. Intrinsic Functions ..
      INTRINSIC        DBLE
*     .. Common blocks ..
      COMMON           /BLOCK1/IPRINT
*     .. Executable Statements ..
      WRITE (NOUT,*)
      WRITE (NOUT,*)
      WRITE (NOUT,*) 'Case 1'
      WRITE (NOUT,*)
*     * Set IPRINT to 1 to obtain output from PRSOL at each iteration *
      IPRINT = 0
      PARAM(1) = 0.2D0
      PARAM(2) = 0.0D0
      N1 = 2
      H = 0.1D0
      PARERR(1) = 1.0D-5
      PARERR(2) = 1.0D-3
      E(1) = 1.0D-4
      E(2) = 1.0D-4
      IFAIL = 1
*
      CALL D02AGF(H,E,PARERR,PARAM,C,N,N1,M1,AUX1,BCAUX1,RNAUX1,PRSOL,
     +            MAT,DUMMY,WSPACE,WSPAC1,WSPAC2,IFAIL)
*
      IF (IFAIL.LT.0) THEN
         WRITE (NOUT,*)
         WRITE (NOUT,99999) ' ** D02AGF returned with IFAIL = ', IFAIL
      ELSE
         IF (IFAIL.EQ.0) THEN
            WRITE (NOUT,*) 'Final parameters'
            WRITE (NOUT,99998) (PARAM(I),I=1,N1)
            WRITE (NOUT,*)
            WRITE (NOUT,*) 'Final solution'
            WRITE (NOUT,*) 'X-value     Components of solution'
            CALL RNAUX1(X,X1,R,PARAM)
            H = (X1-X)/5.0D0
            DO 20 I = 1, 6
               DUM = X + DBLE(I-1)*H
               WRITE (NOUT,99997) DUM, (C(I,J),J=1,N)
   20       CONTINUE
         ELSE
            WRITE (NOUT,99999) 'IFAIL = ', IFAIL
         END IF
      END IF
      RETURN
*
99999 FORMAT (1X,A,I5)
99998 FORMAT (1X,3E16.6)
99997 FORMAT (1X,F7.2,3E13.4)
      END
*
      SUBROUTINE AUX1(F,Y,X,PARAM)
```

```
*     .. Scalar Arguments ..
      DOUBLE PRECISION X
*     .. Array Arguments ..
      DOUBLE PRECISION F(2), PARAM(2), Y(2)
*     .. Executable Statements ..
      F(1) = Y(2)
      F(2) = (Y(1)**3-Y(2))/(2.0D0*X)
      RETURN
      END
*
      SUBROUTINE RNAUX1(X,X1,R,PARAM)
*     .. Scalar Arguments ..
      DOUBLE PRECISION R, X, X1
*     .. Array Arguments ..
      DOUBLE PRECISION PARAM(2)
*     .. Executable Statements ..
      X = 0.1D0
      X1 = 16.0D0
      R = 16.0D0
      RETURN
      END
*
      SUBROUTINE BCAUX1(G,G1,PARAM)
*     .. Array Arguments ..
      DOUBLE PRECISION G(2), G1(2), PARAM(2)
*     .. Local Scalars ..
      DOUBLE PRECISION Z
*     .. Intrinsic Functions ..
      INTRINSIC        SQRT
*     .. Executable Statements ..
      Z = 0.1D0
      G(1) = 0.1D0 + PARAM(1)*SQRT(Z)*0.1D0 + 0.01D0*Z
      G(2) = PARAM(1)*0.05D0/SQRT(Z) + 0.01D0
      G1(1) = 1.0D0/6.0D0
      G1(2) = PARAM(2)
      RETURN
      END
*
      SUBROUTINE EX2
*     .. Parameters ..
      INTEGER          N, M1
      PARAMETER        (N=3,M1=6)
      INTEGER          NOUT
      PARAMETER        (NOUT=6)
*     .. Scalars in Common ..
      INTEGER          IPRINT
*     .. Local Scalars ..
      DOUBLE PRECISION DUM, H, R, X, X1
      INTEGER          I, IFAIL, J
*     .. Local Arrays ..
      DOUBLE PRECISION C(M1,N), DUMMY(1,1), E(N), MAT(N,N), PARAM(N),
     +                 PARERR(N), WSPAC1(N), WSPAC2(N), WSPACE(N,9)
*     .. External Subroutines ..
      EXTERNAL         AUX2, BCAUX2, D02AGF, PRSOL, RNAUX2
*     .. Intrinsic Functions ..
      INTRINSIC        DBLE
*     .. Common blocks ..
      COMMON           /BLOCK1/IPRINT
*     .. Executable Statements ..
      WRITE (NOUT,*)
      WRITE (NOUT,*)
      WRITE (NOUT,*) 'Case 2'
      WRITE (NOUT,*)
*     * Set IPRINT to 1 to obtain output from PRSOL at each iteration *
      IPRINT = 0
      H = 10.0D0
      PARAM(1) = 32.0D0
      PARAM(2) = 6000.0D0
      PARAM(3) = 0.54D0
      PARERR(1) = 1.0D-5
      PARERR(2) = 1.0D-4
```

```
      PARERR(3) = 1.0D-4
      E(1) = 1.0D-2
      E(2) = 1.0D-2
      E(3) = 1.0D-2
      IFAIL = 1
*
      CALL D02AGF(H,E,PARERR,PARAM,C,N,N,M1,AUX2,BCAUX2,RNAUX2,PRSOL,
     +            MAT,DUMMY,WSPACE,WSPAC1,WSPAC2,IFAIL)
*
      IF (IFAIL.LT.0) THEN
         WRITE (NOUT,*)
         WRITE (NOUT,99999) ' ** D02AGF returned with IFAIL = ', IFAIL
      ELSE
         IF (IFAIL.EQ.0) THEN
            WRITE (NOUT,*) 'Final parameters'
            WRITE (NOUT,99998) (PARAM(I),I=1,N)
            WRITE (NOUT,*)
            WRITE (NOUT,*) 'Final solution'
            WRITE (NOUT,*) 'X-value     Components of solution'
            CALL RNAUX2(X,X1,R,PARAM)
            H = (X1-X)/5.0D0
            DO 20 I = 1, 6
               DUM = X + DBLE(I-1)*H
               WRITE (NOUT,99997) DUM, (C(I,J),J=1,N)
   20       CONTINUE
         ELSE
            WRITE (NOUT,99999) 'IFAIL = ', IFAIL
         END IF
      END IF
      RETURN
*
99999 FORMAT (1X,A,I5)
99998 FORMAT (1X,3E16.6)
99997 FORMAT (1X,F7.0,3E13.4)
      END
*
      SUBROUTINE AUX2(F,Y,X,PARAM)
*     .. Scalar Arguments ..
      DOUBLE PRECISION X
*     .. Array Arguments ..
      DOUBLE PRECISION F(3), PARAM(3), Y(3)
*     .. Local Scalars ..
      DOUBLE PRECISION C, S
*     .. Intrinsic Functions ..
      INTRINSIC        COS, SIN
*     .. Executable Statements ..
      C = COS(Y(3))
      S = SIN(Y(3))
      F(1) = S/C
      F(2) = -(PARAM(1)*S+0.00002D0*Y(2)*Y(2))/(Y(2)*C)
      F(3) = -PARAM(1)/(Y(2)*Y(2))
      RETURN
      END
*
      SUBROUTINE RNAUX2(X,X1,R,PARAM)
*     .. Scalar Arguments ..
      DOUBLE PRECISION R, X, X1
*     .. Array Arguments ..
      DOUBLE PRECISION PARAM(3)
*     .. Executable Statements ..
      X = 0.0D0
      X1 = PARAM(2)
      R = PARAM(2)
      RETURN
      END
*
      SUBROUTINE BCAUX2(G,G1,PARAM)
*     .. Array Arguments ..
      DOUBLE PRECISION G(3), G1(3), PARAM(3)
*     .. Executable Statements ..
      G(1) = 0.0D0
```

```
      G(2) = 500.0D0
      G(3) = 0.5D0
      G1(1) = 0.0D0
      G1(2) = 450.0D0
      G1(3) = PARAM(3)
      RETURN
      END
*
      SUBROUTINE PRSOL(PARAM,RESID,N1,ERR)
*     .. Parameters ..
      INTEGER          NOUT
      PARAMETER        (NOUT=6)
*     .. Scalar Arguments ..
      DOUBLE PRECISION RESID
      INTEGER          N1
*     .. Array Arguments ..
      DOUBLE PRECISION ERR(N1), PARAM(N1)
*     .. Scalars in Common ..
      INTEGER          IPRINT
*     .. Local Scalars ..
      INTEGER          I
*     .. Common blocks ..
      COMMON           /BLOCK1/IPRINT
*     .. Executable Statements ..
      IF (IPRINT.NE.0) THEN
         WRITE (NOUT,99999) 'Current parameters   ', (PARAM(I),I=1,N1)
         WRITE (NOUT,99998) 'Residuals   ', (ERR(I),I=1,N1)
         WRITE (NOUT,99998) 'Sum of residuals squared  ', RESID
         WRITE (NOUT,*)
      END IF
      RETURN
*
99999 FORMAT (1X,A,6(E14.6,2X))
99998 FORMAT (1X,A,6(E12.4,1X))
      END
```

9.2 Program Data

None.

9.3 Program Results

```
 D02AGF Example Program Results


 Case 1

 Final parameters
     0.464269E-01    0.349429E-02

 Final solution
 X-value     Components of solution
    0.10   0.1025E+00   0.1734E-01
    3.28   0.1217E+00   0.4180E-02
    6.46   0.1338E+00   0.3576E-02
    9.64   0.1449E+00   0.3418E-02
   12.82   0.1557E+00   0.3414E-02
   16.00   0.1667E+00   0.3494E-02


 Case 2

 Final parameters
     0.323729E+02    0.596317E+04   -0.535231E+00

 Final solution
 X-value     Components of solution
      0.   0.0000E+00   0.5000E+03   0.5000E+00
```

```
1193.   0.5298E+03   0.4516E+03   0.3281E+00
2385.   0.8076E+03   0.4203E+03   0.1231E+00
3578.   0.8208E+03   0.4094E+03  -0.1032E+00
4771.   0.5563E+03   0.4200E+03  -0.3296E+00
5963.   0.0000E+00   0.4500E+03  -0.5352E+00
```